
1.Задание:
Необходимо написать программу, реализующую параллельную работу нескольких процессов. Каждый процесс может состоять из одного или нескольких потоков. Любой из потоков, работающих в составе этих процессов, может быть приостановлен и вновь запущен некоторой определенной клавишей (можно использовать буквенные или цифровые клавиши). 
Вариант№2.
ЗАДАЧИ:

1. Два потока: первый читает информацию из файла (например, стихи или текст программы) в буфер, второй эту информацию из буфера выдаёт на экран. При заполнении окна вывода до конца его содержимое не должно обновляться полностью – вывод новой информации должен осуществляться в последнюю строку, а все остальные строки смещаться вверх (по материалам лаб. работы №4). Имя читаемого файла задавать как параметр командной строки. После окончания файла он начинает считываться заново.

2. Музыкальное сопровождение – как минимум несколько нот различной длительности, образующие мелодию. Во включённом состоянии мелодия должна играть в фоновом режиме (по материалам лаб. работы №2).
5. Вывод заголовка работы (фамилия и имя автора и название работы) в виде “бегущей строки” (по материалам лаб. работы №5). Место вывода (верхняя или нижняя строка экрана) задавать параметром командной строки.
7. Движущийся объект (в простейшем случае, например, летающий и отражающийся от границ окна шарик). Движение может осуществляться свободно или при управлении с клавиатуры (по материалам лаб. работы №5).

Придется мне скриншот Вашей программы помещать…

А теперь давайте искать различия.  Сколько найдете – такова и оценка будет (
Ой, боюсь, она даже до «двойки» не дотянет (
1. Добавьте вывод на эуран информации о величине выделенного потокам кванта времени.

2. Добавьте еще один параметр командной строки – текст «бегущей строки». Если его нет – выводится то, что у Вас сейчас. Если он задан – выводится тот текст, который будет указан при запуске программы. 

3. Присылайте работу только после выполнения задания. 

2.Описание программы

Программа реализует паралелльную работу четырех процессов: процесс 1 «Писатель-Читатель», процесс 2 «Мелодия», процесс3 «Бегущий заголовок», процесс 4 «Летающий объект».
В составе процесса1 работают потоки «Писатель» и «Читатель». Процессы 2, 3 и  4 состоят из одного потока каждый.
Потоки, работающие в составе общего для них процесса могут находится в четырех состояниях: «готов», «приостановлен», «активен», «ожидает».

Самостоятельные потоки могут находится в двух состояниях: «активен» и «приостановлен».

Переключение между состояниями «готов»/«приостановлен» выполняет пользователь с помощью ключевых(управлющих) клавиш. 
В состояние «активен» поток переходит когда ему передается управление и выделяется квант времени.  
В состоянии «готов» поток переходит когда заканчивается выделенный ему квант времени и все действия потока успешно выполнились.

В состояние «ожидает» поток переходит при попытке выполнить недоступное действие. Для потока «Писатель» это попытка записать данные в полностью заполненный буфер. Для потока «Читатель» - попытка прочитать данные из пустого буфера. 
Из состояния ожидания поток выйдет, когда выполнения действия, приведшего к блокировке, станет возможно.
Выбор активного потока в составе процесса может быть случайным или определяться явным образом, в том случае когда один поток заблокирован и ждет выполнения действий другим потоком.

Процессы 2, 3 и 4 выполняются равномерно, через заданное количество «тиков» системного таймера.

 
После запуска программы потоки процесса 1 находятся в состоянии «готов». Потоки процессов 2, 3 и 4 – в состоянии «активен».

Каждому потоку в составе процесса 1 выделяется одинаковый квант времени равный  5 тикам системного таймера. Для каждого потока этот квант можно уменьшить или увеличить при помощи соответствующих управляющих клавиш. Кванты других потоков при этом не изменяются.
Процесс «Мелодия» в фоновом режиме проигрывает 7 нот (одна октава) с разной продолжительностью.

Рабочий экран программы делится на два окна: окно процесса1 и информационное окно.
В окне процесса 1 поток-«читатель» выводит содержимое файла, прочитанное из буфера. Оба потока этого процесса обновляют информацию о проценте заполненности буфера и его текущем содержимом.
В информационном окне выводится:
· Название потока, который выполняется в данный момент и величина выделенного этому потоку кванта времени.

· Список всех потоков системы с указанием текущего статуса.

· Для самостоятельных потоков («Бегущий заголовок» и «Летающий объект») помимо  дополнительно в скобках указано количество тиков, через которое потоку предоставляется управление. Чем меньше это количество, тем быстрее скорость потока.
· Информация об управляющих клавишах.

При запуске программы указывается три параметра.

1. Имя файла – обязательный параметр, строка. 

Имя файла, из которого поток-«писатель» считывает информацию для последующей записи в буфер.
Если указаный файл не найден, программа попытается открыть тестовый файл “kurs.pas”, находящий в одном каталоге с исполняемым файлом. Если и это не получиться, то программа выдаст соответствующее предупреждение и завершится.
2. Положение заголовка – необязательный параметр. 

Если параметр равен 0, заголовок будет отображаться в верхней строке экрана. Если указано любое другое значение или параметр задан, заголовок будет отображаться в нижней строке экрана.
3. Величина задержки – необязательный параметр.

Для возможности пронаблюдать работу системы в замедленном режиме, в каждом из потоков установлена дополнительная задержка (delay), величина этой задержки задается третьим параметром. Если параметр не указан или указан не верно, задержка устанавливается равной 1 секунда.
Управляющие клавиши

	Поток
	Замедление
	Ускорение
	Старт/

приостановка
	Управление движением

	Писатель
	Q
	A
	Z
	

	Читатель
	W
	S
	X
	

	Мелодия
	
	
	C
	

	Бегущий заголовок
	Shift+1
	Shift+2
	Shift+0
	

	Летающий объект
	Ctrl+1
	Ctrl+2
	Ctrl+0
	Ctrl+стрелки


3. Описание основных переменных
Типы:

video_cell=record {описание знакоместа текстового экрана}

   symbol: char;

   attr: byte;

end;

video = array[1..25,1..80] of video_cell;{структура видеопамяти}

type_melody=record {структура одной ноты}

   tone:word; {частоты ноты}

   time:integer; {продолжительность звучания ноты}

end;
Переменные:

memory: video  absolute $B800:$0000; {массив для прямого обращения к видеопамяти}

  save_memory,  save_memory_start:video; {массив для сохранения состояния экрана}

  save_cell:video_cell;

  OldTextAttr:byte;{для сохранения текущих атрибутов экрана}

{Процесс 1: Поток1 - Писатель, Поток2 - Читатель} 

  kvant_time1,kvant_time2:word; {квант времени, выделенное потоку}

  potok_time:word; {время, уже отработанное потоком}

  writer_activ,reader_activ:boolean; {признак активности потока: true-активен, false-приостановлен}

  writer_wait,reader_wait:boolean;{признак статуса ожидания}

  pos_writer:byte; {индекс следующего записываемого элемента}

  pos_reader:byte; {индекс следующего считываемого элемента}

  select_potok:byte; {номер активного потока в процессе Писатель-Читатель}

  zapusk_potoka:byte;{номер потока, который надо принудительно запустить} 

  name_file: text; {имя файла}

  s, s_n: string; {строки, прочитанные из файла}

  bufer:array[1..7] of string; {буфер для хранения строк, прочитанных из файла}

  bufer_size:byte;

  pos_window:byte; {позиция для вывода строки из буфера}

  x_last,y_last: integer; {для сохранения положения курсора}

{Процесс 2: Музыка} 

  melody:array[1..8] of type_melody; {мелодия из 8-ми нот }

  work_time_snd:integer; {продолжительность звучания текущей ноты}

  time_del_snd:integer; {заданное время звучания ноты}

  nom_note:byte; {номер ноты}

  sound_activ:boolean; {включение/выключение музыки}

{Процесс 3: Бегущий заголовок}

  fio:string;{строка заголовка}

  len_str:integer; {длина бегущей строки}  

  work_time_fio:integer; {время отображения заголовка в текущей позиции}

  time_del_fio:integer; {заданное время отображения заголовка}

  fio_activ:boolean; {остановка/запуск движения заголовка}

  y_now_fio,y_next_fio,y_work:integer; {координаты положения заголовка}

  pos_zagolovka:byte;

{Процесс 4: Летающий объект}

  work_time_fly:integer; {время отображения объекта в текущем месте экрана}

  time_del_fly:integer; {заданная скорость движения}

  fly_activ:boolean; {остановка/запуск летающего объекта }

  x_now,x_next,y_now,y_next:integer; {координаты положения объекта}

  x_old,y_old: integer; {для сохранения положения курсора}

  way, old_way:byte;{текущее направление движения объекта}

  sym:char;{символ для отображения объекта}

  prepare_exit:boolean; {флаг подготовки к завершению работы}

  end_work:boolean; {признак завершения работы процессе Писатель-Читатель}

  period_del:word; {время задержки}

  c:byte;{символ нажатой клавиши}

  Old_Key: Procedure; {используется для сохранения стандартного обработчика прерывания системного таймера}

  Old_Tic: Procedure;{используется для сохранения стандартного обработчика прерывания клавиатуры}

4. Описание основных блоков (подпрограмм) программы;
Procedure Reader Writer - управляющий блок для процесса 1: Писатель-Читатель Получив управление, процедура запускает на выполнение поток-писатель или поток-читатель. Выбор потока происходит случайным образом или запускается тот поток, вызов которого был затребован другим потоком.

Procedure Writer– поток-писатель из процесса1. 
Производит построчное чтение информации из файла и записывает её в буфер. Достигнув конца файла, начинается считывание заново.
Procedure Reader – поток-читатель из процесса1. 
Выбирает информацию из буфера и выводит её на экран. Чтение строк из буфера происходит по принципу «очереди»(FIFO)

Procedure New_Key – процедура-обработчик прерывания от клавиатуры
Проверяет код нажатых клавиш клавиатуры. Если нажата управляющая клавиша, то передает соответствующему потоку информацию о действиях, которые поток должен выполнить (замедление/ускорение/остановка)
Procedure New_Tic – процедура-обработчик прерывания от системного таймера.
Для процесса «Мелодия» начинает и завершает проигрывание ноты заданной частоты и продолжительности.

Для процесса «Летающий объект» по истечении определенного кванта времени прорисовывает объект в новом местоположении экрана.
Для процесса «Бегущий заголовок» по истечении определенного кванта времени выводит заголовок в новом местоположении экрана.

Procedure Start_Init – прорисовывает рабочее окно программы, выводит на экран справочную информацию, выполняет начальную инициализацию процессов и потоков системы.

5. Текст программы
Program Kurs;

{$M $1000,0,0}

uses crt,dos;

type video_cell=record

   symbol: char;

   attr: byte;

end;

type type_melody=record {структура одной ноты}

   tone:word; {частоты ноты}

   time:integer; {продолжительность звучания ноты}

end; 

type video = array[1..25,1..80] of video_cell;{структура видеопамяти}

var

  memory: video  absolute $B800:$0000; {массив для прямого обращения к видеопамяти}

  save_memory,  save_memory_start:video; {массив для сохранения состояния экрана}

  save_cell:video_cell;

  OldTextAttr:byte;{для сохранения текущих атрибутов экрана}

{Процесс 1: Поток1 - Писатель, Поток2 - Читатель} 

  kvant_time1,kvant_time2:word; {квант времени, выделенное потоку}

  potok_time:word; {время, уже отработанное потоком}

  writer_activ,reader_activ:boolean; {признак активности потока: true-активен, false-приостановлен}

  writer_wait,reader_wait:boolean;{признак статуса ожидания}

  pos_writer:byte; {индекс следующего записываемого элемента}

  pos_reader:byte; {индекс следующего считываемого элемента}

  select_potok:byte; {номер активного потока в процессе Писатель-Читатель}

  zapusk_potoka:byte;{номер потока, который надо принудительно запустить} 

  name_file: text; {имя файла}

  s, s_n: string; {строки, прочитанные из файла}

  bufer:array[1..7] of string; {буфер для хранения строк, прочитанных из файла}

  bufer_size:byte;

  pos_window:byte; {позиция для вывода строки из буфера}

  x_last,y_last: integer; {для сохранения положения курсора}

{Процесс 2: Музыка} 

  melody:array[1..8] of type_melody; {мелодия из 8-ми нот }

  work_time_snd:integer; {продолжительность звучания текущей ноты}

  time_del_snd:integer; {заданное время звучания ноты}

  nom_note:byte; {номер ноты}

  sound_activ:boolean; {включение/выключение музыки}

{Процесс 3: Бегущий заголовок}

  fio:string;{строка заголовка}

  len_str:integer; {длина бегущей строки}  

  work_time_fio:integer; {время отображения заголовка в текущей позиции}

  time_del_fio:integer; {заданное время отображения заголовка}

  fio_activ:boolean; {остановка/запуск движения заголовка}

  y_now_fio,y_next_fio,y_work:integer; {координаты положения заголовка}

  pos_zagolovka:byte;

{Процесс 4: Летающий объект}

  work_time_fly:integer; {время отображения объекта в текущем месте экрана}

  time_del_fly:integer; {заданная скорость движения}

  fly_activ:boolean; {остановка/запуск летающего объекта }

  x_now,x_next,y_now,y_next:integer; {координаты положения объекта}

  x_old,y_old: integer; {для сохранения положения курсора}

  way, old_way:byte;{текущее направление движения объекта}

  sym:char;{символ для отображения объекта}

  prepare_exit:boolean; {флаг подготовки к завершению работы}

  end_work:boolean; {признак завершения работы процессе Писатель-Читатель}

  period_del:word; {время задержки}

  i:integer;

  c:byte;{символ нажатой клавиши}

  Old_Key: Procedure; {используется для сохранения стандартного обработчика прерывания системного таймера}

  Old_Tic: Procedure;{используется для сохранения стандартного обработчика прерывания клавиатуры}

Var Regs:Registers;

{$F+}

Procedure New_Key; Interrupt;

Begin

 c:=Port[$60];{получаем код нажатой клавиши}

 if c=1 then  {подготовка к завершению программы }

   prepare_exit:=true

  else

    begin 

     case c of

      {управление процессом Писатель-Читатель }     

        16:begin {уменьшение кванта времени для писателя}

            kvant_time1:=kvant_time1-1;

            if kvant_time1<1 then kvant_time1:=1;

           end;         

        30:kvant_time1:=kvant_time1+1;      {увеличение кванта времени для писателя}   


    44:begin

             writer_activ:=not writer_activ;{включение/выключение писателя}                  

             if not writer_activ then

               begin 

                 gotoxy(4,19);        

                 write('Писатель: приостановлен ');

               end

             else

               begin

                 gotoxy(4,19);        

                 write('Писатель: готов        ');

               end;                                 

            end;  

        17: begin {уменьшение кванта времени для читателя}

              kvant_time2:=kvant_time2-1;

              if kvant_time2<1 then kvant_time2:=1;

            end;    

        31:kvant_time2:=kvant_time2+1; {увеличение кванта времени для читателя}                

        45: begin 

                reader_activ:=not reader_activ;{включение/выключение читателя}

                if not reader_activ then

                  begin 

                    gotoxy(4,20);        

                    write('Читатель: приостановлен   ');

                  end

                else

                begin

                    gotoxy(4,20);        

                    write('Читатель: готов         ');

                end;               

            end;   


    46:begin


     sound_activ:=not sound_activ;{включение/выключение писателя}


     if not sound_activ then

               begin 




     nosound;

                 gotoxy(4,21);        

                 write('Мелодия: приостановлен ');

               end

             else

               begin

                 gotoxy(4,21);        

                 write('Мелодия: активен        ');

               end;                                 

            end;  



        end; {case}

        {управление процессом 4-движущийся объект}

        If ((mem[$40:$17] and $4) =$4) then {если нажата клавиша CTRL}

             case c of {меняем направление движения}

                72: way:=1;

                80: way:=2;

                75: way:=3;

                77: way:=4;

                11: begin {клавиша 0:меняем статус потока=приостановлен/активен}

                      fly_activ:=not fly_activ;    

                       if fly_activ then

                         begin

                           way:=old_way;

                           gotoxy(4,23);

                           write('Движение объекта: активен (',time_del_fly,')    ');  

                         end

                      else

                         begin

                           old_way:=way;

                           way:=0;

                           gotoxy(4,23);

                           write('Движение объекта: приостановлен');                             

                         end;   

                    end;   

                130: begin

                      {увеличиваем время отображения объекта в тек.позиции- замедление движения}                

                       time_del_fly:=time_del_fly+1; 

                       if fly_activ then 

                          begin

                            gotoxy(4,23);

                            write('Движение объекта: активен (',time_del_fly,')     ');  

                          end;    

                     end;                        

                131: 

                     begin 

                      {уменьшаем время отображения объекта в тек.позиции- - ускорение движения}                     

                       time_del_fly:=time_del_fly-1; 

                       if time_del_fly<1 then time_del_fly:=1;

                       if fly_activ then 

                        begin

                           gotoxy(4,23);

                           write('Движение объекта: активен (',time_del_fly,')     ');  

                        end;                                             

                     end;

                end; {case}

            {управление процессом 3 - бегущий заголовок}

        If (((mem[$40:$17] and $2) =$2) or ((mem[$40:$17] and $1)=$1)) then {если нажата клавиша SHIFT}

              case c of

                11: begin

                      fio_activ:=not fio_activ; {клавиша 0:меняем статус потока=приостановлен/активен}

                      gotoxy(4,22);

                      if fio_activ then             

                         write('Заголовок: активен (',time_del_fio,')     ')             

                      else

                         write('Заголовок: приостановлен'); 

                   end;    

                130: begin

                {клавиша 1: увеличиваем время отображения заголовка в тек.позиции - замедление движения}   

                        time_del_fio:=time_del_fio+1;      

                        gotoxy(4,22);

                        write('Заголовок: активен (',time_del_fio,')     ');    

                     end;  

                131: begin 

                {клавиша 2: увеличиваем время отображения заголовка в тек.позиции - ускорение движения}

                        time_del_fio:=time_del_fio-1;

                        if time_del_fio<1 then time_del_fio:=1;

                        gotoxy(4,22);

                        write('Заголовок: активен (',time_del_fio,')     ');              

                     end;

               end;

  end; {не ESC}

 if ((not fio_activ) and (not fly_activ) and (not sound_activ)

     and (not writer_activ) and (not reader_activ) ) then

  {если все потоки остановлены, устанавливаем флаг подготовки к завершению программы}

       prepare_exit:=true;

  Inline($9C);

  Old_Key;

End;

Procedure New_Tic;Interrupt;

begin

  inc(work_time_snd);

  inc(work_time_fly);

  inc(work_time_fio);

  inc(potok_time);

{работа процесса 2 -мелодия} 

   if work_time_snd>=time_del_snd then

    if sound_activ then {если звук включен}

     begin

       nom_note:=nom_note+1;

       if nom_note>5 then nom_note:=1;

       work_time_snd:=0;


   time_del_snd:=melody[nom_note].time;


   nosound;


   sound(melody[nom_note].tone);

   end;

{работа процесса 3 -бегущий заголовок}

  if work_time_fio>=time_del_fio then

  begin

      if fio_activ then {если флаг остановки не установлен}

        begin


      for i:=0 to (len_str-1) do


      begin



    y_work:=y_now_fio+i;




if y_work>80 then y_work:=y_work-80;



    memory[pos_zagolovka][y_work].symbol:=#32;


      end;




      y_next_fio:=y_now_fio-1;

          if y_next_fio<1 then y_next_fio:=80-y_next_fio;


      for i:=0 to (len_str-1) do


      begin



    y_work:=y_next_fio+i;




if y_work>80 then y_work:=y_work-80;


        memory[pos_zagolovka][y_work].symbol:=fio[i+1]; {отображаем заголовок в следующем положении}




memory[pos_zagolovka][y_work].attr:=14; {устанавливаем цвет заголовка}

          end;

  
     {сдвигаем текущие координаты}


      y_now_fio:=y_next_fio;


      work_time_fio:=0; {обнуляем счетчик времени}

        end

  end;

 {работа потока 4 - летающий объект}

   if work_time_fly>=time_del_fly then

      begin

         work_time_fly:=0; {обнуляем счетчик времени}     

         if not fly_activ then {если поток не активен}

            begin {отображаем остановившийся объект}

               memory[x_now][y_now].symbol:=sym;

               memory[x_now][y_now].attr:=(memory[x_next][y_next].attr div 16)*16+12;       

            end

         else

            begin

               case way of {в зависимости от направления движения, определяем координату следующего положения объекта}

                  1: x_next:=x_now-1; 

                  2: x_next:=x_now+1; 

                  3: y_next:=y_now-1;

                  4: y_next:=y_now+1; 

               end;

              {проверяем не вышел ли объект за границы экрана}

              if y_next>80 then y_next:=1;

              if y_next<1 then y_next:=80;

              if x_next>24 then x_next:=2;

              if x_next<2 then x_next:=24;

              {воостанавливаем состояние экрана в текущем положении объекта}

              memory[x_now][y_now]:=save_memory[x_now][y_now];

              {сохраняем состояние экрана в следующем местоположении объекта}

              save_memory[x_next][y_next]:=memory[x_next][y_next];

              memory[x_next][y_next].symbol:=sym; {отображаем движущийся объект в следующем положении}

              memory[x_next][y_next].attr:=(memory[x_next][y_next].attr div 16)*16+12;

              x_now:=x_next; {сдвигаем текущие координаты}

              y_now:=y_next;      

             end;


 end;

  inline($9C);

  Old_tic;

end;

{$F-}

Procedure  SetCurSize(BegLine, EndLine: Byte);

Var Regs:Registers;

Begin

    With Regs Do

    Begin

       AH:=$01;

       CH:=BegLine;

       CL:=EndLine;

      End;

    Intr($10,Regs);

End;

procedure Writer;

Begin

    GotoXY(2,15);

    Write('Выполняется: Писатель     ');

    gotoxy(2,16);

    Write('Выделен квант =',kvant_time1,'  ' );

    potok_time:=0;

    zapusk_potoka:=0;

    if not writer_activ then {1.Писатель приостановлен}

       begin

         gotoxy(4,19);        

         write('Писатель: приостановлен          ');

         delay(period_del);

        end

    else {1.Писатель активен}

      begin

        while ((potok_time<kvant_time1) and (not prepare_exit)) do {пока не закончился квант времени }

         {и не установлен флаг завершения работы} 

          if ((pos_writer=bufer_size) and (pos_reader=1)) or (pos_reader=pos_writer+1) then

            {3.в буфере нет свободных ячеек}

            begin

              textcolor(lightred);

              gotoxy(14,13);

              write('переполнен');

              textcolor(white);

              gotoxy(4,19);        

              write('Писатель: ожидает        ');

              writer_wait:=true; {переводим писателя в состояние ожидания и завершаем поток}

              if  reader_activ then  {читатель активен}

                 zapusk_potoka:=2 {инициируем запуск читателя}

              else

                 zapusk_potoka:=0;

              delay(period_del);

              break;  

            end

            else {в буфере есть свободные ячейки}        

               begin

                 writer_wait:=false; 

                 gotoxy(4,19);        

                 write('Писатель: активен          ');

                 gotoxy(14,13);

                 write('           ');

                 if eof(name_file) then {если достигнут конец файла, начинаем чтение сначала}

                     reset(name_file)

                 else

                    begin

                      readln(name_file,s);

                      bufer[pos_writer]:=s; {помещаем строку в буфер}

                      inc(pos_writer); {увеличиваем количество заполненных ячеек буфера}

                      gotoxy(8,13);

                      if pos_writer>bufer_size then

                        pos_writer:=1;

                      if pos_writer<pos_reader then

                        write(round((pos_writer-pos_reader+7)*100/(bufer_size-1)),'%  ')


                      else

                       write(round((pos_writer-pos_reader)*100/(bufer_size-1)),'%  ');

                      delay(period_del);                  

                    end

               end;

          if (writer_activ and (not writer_wait)) then

            begin

              gotoxy(4,19);        

              write('Писатель: готов        ');          

            end;    

      end;

End;   

Procedure Reader;

var k,l,count_str:integer;

begin

 if prepare_exit then {флаг завершения работы, дочитываем информацию из буфера}

   begin

     while not (pos_reader=pos_writer) do

       begin       

         reader_wait:=false;

         s_n:=Bufer[pos_reader];

         pos_reader:=pos_reader+1;

         window(2,5,78,12); 

         gotoxy(x_last,y_last);


     save_cell:=memory[x_now][y_now];



 memory[x_now][y_now]:=save_memory[x_now][y_now];;

         writeln(s_n); { выводим строку на экран}



 save_memory:=memory;



 memory[x_now][y_now]:=save_cell;

         x_last:=whereX;

         y_last:=whereY;

         window(1,1,80,25);

         if pos_reader>bufer_size then

           pos_reader:=1;               

          gotoxy(2,13);

         if pos_writer<pos_reader then


        write(round((pos_writer-pos_reader+7)*100/(bufer_size-1)),'%  ')


      else


        write(round((pos_writer-pos_reader)*100/(bufer_size-1)),'%  ');

         delay(period_del);

        end;

       reset(name_file);

       close(name_file);      

       end_work:=true;

       delay(period_del);

    end

 else 

    begin

       GotoXY(2,15);

       Write('Выполняется: Читатель     ');

       gotoxy(2,16);

       Write('Выделен квант =',kvant_time2,'  ' );

       potok_time:=0;

       zapusk_potoka:=0;

       if not reader_activ then {1.читатель приостановлен}

          begin

            gotoxy(4,20);        

            write('Читатель: приостановлен');

            delay(period_del);

          end      

       else

         begin

           while ((potok_time<kvant_time2)  and (not prepare_exit))  do {пока не закончился квант времени }

             {и не установлен флаг завершения работы}

                if pos_reader=pos_writer then {3.буфер пуст} 

                   begin

                     textcolor(lightred);

                     gotoxy(14,13);

                     write('пуст                 ');

                     textcolor(white);

                     textcolor(white);

                     gotoxy(4,20);        

                     write('Читатель: ожидает       ');

                     reader_wait:=true;

                     if writer_activ then

                       zapusk_potoka:=1 {инициируем запуск писателя}

                     else

                       zapusk_potoka:=0;

                     delay(period_del);

                     break;

                   end

                else                  

                   begin

                      gotoxy(4,20);        

                      write('Читатель: активен     ');

                      gotoxy(14,13);

                      write('           ');

                      reader_wait:=false;

                      s_n:=Bufer[pos_reader];

                      pos_reader:=pos_reader+1;                   

                      window(2,5,78,12); 

                      gotoxy(x_last,y_last);






  save_cell:=memory[x_now][y_now];






  memory[x_now][y_now]:=save_memory[x_now][y_now];;

                      writeln(s_n); { выводим строку на экран}






  save_memory:=memory;






  memory[x_now][y_now]:=save_cell;

                      x_last:=whereX;

                      y_last:=whereY;

                      window(1,1,80,25);

                      if pos_reader>bufer_size then

                         pos_reader:=1;                 

                      gotoxy(8,13);

                      if pos_writer<pos_reader then

                         write(round((pos_writer-pos_reader+7)*100/(bufer_size-1)),'%  ')

                      else

                         write(round((pos_writer-pos_reader)*100/(bufer_size-1)),'%  ');

                      delay(period_del);                  

                  end; 

             if (reader_activ and (not reader_wait)) then

               begin

                 gotoxy(4,20);        

                 write('Читатель: готов        ');          

               end;                          

          end

    end

End;         

{управляющий блок для процесса: Писатель-Читатель}     

procedure ReaderWriter;

begin

  if zapusk_potoka>0 then

    {если один поток инициировал вызов другого потока }

     select_potok:=zapusk_potoka

  else {иначе поток выбираем случайным образом}

     select_potok:=random(2)+1;

  if (select_potok=1) then 

     Writer;

  if (select_potok=2) then

     Reader;

end;

{Начальное заполнение переменных, прорисовка экрана, вывод справочной информации}

procedure Start_Init;

Begin

    clrscr;

    {начальная инициализация переменных}   



melody[1].tone:=261;  {до}


melody[1].time:=6;


melody[2].tone:=294;  {ре}


melody[2].time:=5;


melody[3].tone:=330;  {ми}


melody[3].time:=5;


melody[4].tone:=349;  {фа}


melody[4].time:=5;



melody[5].tone:=392;  {соль}


melody[5].time:=5;


melody[5].tone:=440;  {ля}


melody[5].time:=5;


melody[5].tone:=494;  {си}


melody[5].time:=5;


melody[5].tone:=523;  {до}


melody[5].time:=6;


sound_activ:=true;


nom_note:=1;


work_time_snd:=0;


time_del_snd:=melody[1].time;


sound(melody[1].tone);

    time_del_fly:=4;

    work_time_fly:=0;

    time_del_fio:=5;

    work_time_fio:=0;

    fio_activ:=true;


fio:='Сытник Владимир Александрович Курсовая работа: Параллельная работа процессов ';

    len_str:=length(fio);


y_now_fio:=2;

    fly_activ:=true;

    writer_activ:=true;

    writer_wait:=false;

    reader_activ:=true;

    reader_wait:=false;

    {объект начнет движение вправо}

    x_now:=14;

    y_now:=40;

    x_next:=x_now;

    y_next:=y_now+1;

    way:=4;    

    x_last:=1;

    y_last:=1;  

    sym:=#4;

    end_work:=false;

    prepare_exit:=false;

    bufer_size:=7;

    pos_writer:=1;

    pos_reader:=bufer_size;

    pos_window:=0;

    kvant_time1:=50;

    kvant_time2:=50;  

    for i:=1 to 79 do

      begin

         memory[2][i].attr:=144;

         memory[14][i].attr:=144; 

         memory[24][i].attr:=144;

      end;

    for i:=2 to 24 do

      begin

         memory[i][1].attr:=144;

         memory[i][80].attr:=144;

      end;

    textcolor(LightGreen);

    gotoxy(37,3);

    write('Процесс1:');

    gotoxy(32,4);

    write('Писатель-читатель');

    textcolor(white);

    gotoxy(2,13);

    write('Буфер:');

    gotoxy(2,15);

    write('Выполняется:');

    gotoxy(2,16);    

    write('Выделен квант=');

    {в момент старта все потоки находятся в состоянии готовности}

    gotoxy(2,18);

    write('Состояние системы:');

    gotoxy(4,19);

    write('Писатель: готов');  

    gotoxy(4,20);

    write('Читатель: готов');   


gotoxy(4,21);

    write('Мелодия: активен');   

    gotoxy(4,23);

    write('Движение объекта: активен (',time_del_fly,')     ');  

    gotoxy(4,22);

    write('Заголовок: активен (',time_del_fio,')     ') ;  

    gotoxy(43,15);

    write('Управляющие клавиши:');      

    gotoxy(40,17);

    write('Замедл|Ускор    Выкл|вкл   Направление'); 


gotoxy(40,19);

    write('   Q  |  A          Z  '); 

    gotoxy(40,20);

    write('   W  |  S          X  '); 


gotoxy(40,21);

    write('      |             C  '); 

    gotoxy(40,23);

    write('Ctrl+1|Ctrl+2     Ctrl+0   Ctrl+стрелки');  

    gotoxy(40,22);

    write('Shft+1|Shft+2     Shft+0  ');    

End;

var

file_error:boolean;

file_str, pos_str,del_str:string;

pos_popytka:longint;

pos_priznak:word;

del_popytka:longint;

del_priznak:word;

BEGIN

    randomize;

    save_memory_start:=memory; {сохраняем начальное состояние экрана}

    OldTextAttr:=TextAttr;{сохраняем текущие атрибуты экрана}

    SetCurSize($20,$0);

    {обработка параметров командной строки} 

     if paramCount=0 then 

       begin

          writeln('KURS имя_файла [положение_заголовка] [величина_задержки]');

          writeln(' имя_файла - строка с именем файла для чтения');    

          writeln(' положение_заголовка - если 0 : заголовок выводится вверху, иначе - внизу');

          writeln(' величина_задержки - время задержки потока в секундах, целое число');

          writeln;

          writeln('Необходимо указать имя файла!');

       end

    else {программа запущена с параметрами}

      begin

          file_error:=false;        

          file_str:=paramStr(1);

          Assign (name_file, file_str);

          {$I-} 

          Reset (name_file); 

          {$I+} 

          if IOResult<>0 then 

             begin {файл открыть не удалось, пытаемся открыть тестовый файл}

                file_str:='kurs.pas';

                Assign (name_file, file_str);

                {$I-} 

                Reset (name_file); 

                {$I+} 

                if IOResult<>0 then 

                   begin

                      file_error:=true;

                      writeln ('Ошибка: не удалось открыть файл данных!! Возможно указано неверное имя файла'); 

                   end;      

              end;

        If not file_error then

           begin

             pos_str:=paramStr(2);

             val(pos_str,pos_popytka,pos_priznak);  

             if ((pos_priznak=0) and (pos_popytka=0)) then

                pos_zagolovka:=1 {заголовок вверху}

             else 

               pos_zagolovka:=25; {заголовок внизу}

             del_str:=paramStr(3);

             val(del_str,del_popytka,del_priznak);  

             if ((del_priznak=0) and (del_popytka>0)) then

                period_del:=del_popytka*1000

             else 

               period_del:=10000;   

             start_init;

             GetIntVec($8,@Old_Tic);{сохраняем старый вектор прерывания системного таймера}

             SetIntVec($8,Addr(New_Tic));{переназначаем его на процедуру обработки прерываний New_Tic}

             GetIntVec($9,@Old_Key);{получаем старый вектор прерывания клавиатуры}

             SetIntVec($9,Addr(New_Key));{переназначаем его на процедуру обработки прерывания New_Key}

             save_memory:=memory;

         while (not end_work) do {пока не процессы1 и 2 не завершили свою работу}

               begin

                 if prepare_exit then

                    begin






  gotoXY(2,15);

                      textcolor(lightred);

                      write('ЗАВЕРШЕНИЕ РАБОТЫ                         ');

                      textcolor(white);

                      reader; 

                    end

                else 

           begin

            ReaderWriter;

           end;

      end;

      SetCurSize($06,$07);{восстанавливаем размеры курсора}

      SetIntVec($9,Addr(Old_key));{восстанавливаем обработчики прерывания}

      SetIntVec($8,Addr(Old_tic));

      TextAttr:=OldTextAttr;{восстанавливаем атрибуты экрана}

      nosound;


  clrscr;

      MemW[Seg0040:$1C]:=MemW[Seg0040:$1A]

     end;

   end;

END.
